
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of EMM test case 9.2.2.2.14 which are part of the LTE test suite.

The test case can be demonstrated to run with two LTE UEs (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Cases:
tc_9_2_2_2_14
Test Group:
EMM
ATS Version:
iwd-EUTRA-TVB2009-12_D10wk49
System Simulator used:
Anite Conformance Toolset
UE(s) used:                       LG L2000 and Huawei Polaris UE
Verification Status:
PASS
4 Corrections required for test case 9.2.2.2.14
4.1 Introduction

This section describes the changes required to make test case 9.2.2.2.14 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D10wk49 release.

4.2 Change 1

	Testcase name
	9.2.2.2.14

	Protocol
	EMM

	Function Name
	function f_TC_9_2_2_2_14_EUTRA()

	Reason for change
	1. Get functions are called before cells initialisation and also PLMN type used in the cells initialisation is LTE_AllCellsOnSamePLMN for which Cell I does not get initialised and Cell I is used in the test case
2. As per prose 36.523-1 as part of the preamble the UE should be in ‘state 3’ on Cell I and T3402 is set to 2 minutes in UE

3. Since the redundant functions f_EUTRA_Preamble () and f_EUTRA_UE_Detach_SwitchOffUe() are removed the AttachRequest either can come with or without security protection

4. The PTI from the AttachRequest is not stored to fill the same in AttachAccept message
5. Default TAI list is filled in AttachAccept message.
6. T3402 timer is an optional IE in the AttachAccept message
7. After SS receives ‘AttachComplete’ RrcConnection need to be released so that the paging message in the next step can be received by UE

	Summary of change
	1. Get functions are called after cells initialisation and also PLMN type LTE_CellsOnDifferentPLMN is used for the cells initialisation which  initialises cell I as well.
2. The redundant functions f_EUTRA_Preamble () and f_EUTRA_UE_Detach_SwitchOffUe() are removed and the steps needed to move UE into ‘state 3’ are added.

3. AttchRequest is handled with or without security protection.

4. PTI from AttachRequest is stored and the same is filled in AttachAccept message
5. TAI list is corrected to of Cell I
6. T3402 timer is corrected to TV format

7. RrcConnecton is released after SS receives AttachComplete message from UE

	Source of change
	File : …\9_2\NAS_Detach.ttcn


Before:
	function f_TC_9_2_2_2_14_EUTRA() runs on EUTRA_PTC

{

   // @desc NW initiated detach / Abnormal case / EMM cause not included

    /* Test Purpose (TP)

     (1)

    with { UE in EMM-REGISTERED state }

    ensure that {

      when { UE receives a DETACH REQUEST message with the Detach type IE "re-attach not required" and with no EMM cause IE included }

        then { UE delete any GUTI, TAI list, last visited registered TAI, list of equivalent PLMNs, KSI, set the update status to EU2 NOT UPDATED and start timer T3402. when T3402 expired, the UE attempts to attach again }

     }

    System Simulator:

    -   Cell I are configured according to Table 6.3.2.2-3 in TS 36.508.

    -   Cell I (VPLMN and set as a Serving cell)

    UE:

    -   The test USIM contains GUTI-9 and TAI-9, and EPS update status is "EU1: UPDATED".

    Preamble:

    -   UE is in state Registered, Idle Mode (state 2) on Cell I according to TS 36.508

    -   T3402 in the UE is set to 2 minutes @sic R5s102201 sic@

    */

  var NAS_MSG_Indication_Type v_NasInd;

  var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(  ); // R5-103673

  var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_CellInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info ( eutra_CellI );

  var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellI);

  var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
  f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c1, LTE_NAS_Usef1_f2);
  // create cell and send out system information

  f_EUTRA_CellConfig_Def(eutra_CellI);

  f_EUTRA_Preamble (eutra_CellI, STATE2_IDLEUPDATE);

  // now to transmit T3402 to UE with value 2 minuts

  f_EUTRA_UE_Detach_SwitchOffUe ( eutra_CellI, RRC_IDLE );
  f_UT_SwitchOnUE (UT, false);

  v_NasInd:= f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellI,

                  tsc_RRC_TI_Def,

                  cr_EstablishmentCause_Any,

                  cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                    cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                    cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                              f_GetPdnType())));

  SRB.send(cas_SRB1_RrcNasPdu_REQ ( eutra_CellI,

                                      cs_TimingInfo_Now,

                                      cs_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est(tsc_RRC_TI_Def,

                                                                                       v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                       v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                       v_AntennaInfo

                                                                                       ),

                                      cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cds_ATTACH_ACCEPT_t3402(v_AttachType,

                                                                                cs_TrackingAreaIdListDef_lv,
                                                                                f_GutiParameters2MobileIdentity(tsc_IEI_Guti, f_EUTRA_CellInfo_GetGuti(eutra_CellI)),

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.aTTACH_REQUEST.addUpdateType),

                                                                                cs_GprsTimer_v('001'B, '00010'B)), //2 minutes
                                                        cs_508_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                                  tsc_PTI_Unassigned,
                                                                                                  cs_AccessPointName(px_AccessPointName),

                                                                                                  f_GetPDNAddress (f_CheckPCOforIPallocationViaNas (omit)),

                                                                                                  omit,

                                                                                                  omit,

                                                                                                  f_GetDefPCOforPCSCFAddress (v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions))))); // @sic R5-104703 sic@

    //Receive RRCConnectionReconfigurationComplete

  SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellI, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

  f_EUTRA_NAS_AttachComplete(eutra_CellI, tsc_EpsDefaultBearerId);

/////////////////////////////////////////////////////////////////

  f_EUTRA_RbEst_Def(eutra_CellI);

  f_EUTRA_TestBody_Set(true);

  fl_TC_9_2_2_2_14_Body();

  // Postamble

  f_EUTRA_TestBody_Set(false);

  f_EUTRA_Postamble(eutra_CellI, E2_CONNECTED);

}


After:
	function f_TC_9_2_2_2_14_EUTRA() runs on EUTRA_PTC

{

   // @desc NW initiated detach / Abnormal case / EMM cause not included

    /* Test Purpose (TP)

     (1)

    with { UE in EMM-REGISTERED state }

    ensure that {

      when { UE receives a DETACH REQUEST message with the Detach type IE "re-attach not required" and with no EMM cause IE included }

        then { UE delete any GUTI, TAI list, last visited registered TAI, list of equivalent PLMNs, KSI, set the update status to EU2 NOT UPDATED and start timer T3402. when T3402 expired, the UE attempts to attach again }

     }

    System Simulator:

    -   Cell I are configured according to Table 6.3.2.2-3 in TS 36.508.

    -   Cell I (VPLMN and set as a Serving cell)

    UE:

    -   The test USIM contains GUTI-9 and TAI-9, and EPS update status is "EU1: UPDATED".

    Preamble:

    -   UE is in state Registered, Idle Mode (state 2) on Cell I according to TS 36.508

    -   T3402 in the UE is set to 2 minutes @sic R5s102201 sic@

    */

  var NAS_MSG_Indication_Type v_NasInd;

  var NAS_AttDetValue_Type v_AttachType; // R5-103673

  var template (omit) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_CellInfo := omit;

  var template (omit) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := omit;

  var template AdditionalUpdateType v_AdditionalUpdateType;

  var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

  var NasCount_Type v_NasCountUL;

  var EUTRA_SecurityParams_Type v_SecurityParams;

  var GutiParameters_Type v_GutiParams;

  var NAS_PlmnId v_PLMN ;

  var TrackingAreaCode v_TAC;

  var template (present) SecurityHeaderType v_ExpectedSecurityHeader;
var template (present) SecurityHeaderType v_AttReqSecurityHeader := (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected);
  f_EUTRA_NAS_Init(LTE_CellsOnDifferentPLMN, c1, LTE_NAS_Usef1_f2);

  v_AttachType := f_GetEAttachType(); // R5-103673

  v_EUTRA_FDD_TDD_CellInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info ( eutra_CellI );

  v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellI);

  v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

  v_SecurityParams := f_EUTRA_Security_Get();

  v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellI);

  v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

  v_TAC := f_EUTRA_CellInfo_GetTAC(eutra_CellI);
  // create cell and send out system information

  f_EUTRA_CellConfig_Def(eutra_CellI);

  //f_EUTRA_Preamble (eutra_CellI, STATE2_IDLEUPDATE);
  // now to transmit T3402 to UE with value 2 minuts

 // f_EUTRA_UE_Detach_SwitchOffUe ( eutra_CellI, RRC_IDLE );
  //f_UT_SwitchOnUE (UT, false);
  f_EUTRA_SetCellPower ( eutra_CellI, tsc_ServingCellRS_EPRE, cs_TimingInfo_Now );

  f_UT_SwitchOnUE (UT, false);

  // Activate DHCP and ICMPv6

  f_DHCPv4_Start(IP, px_IPv4_RemoteAddress, cs_DrbInfo_EUTRA(eutra_CellI, tsc_DRB1));   /* @sic R5s100135, R5s100189 sic@ */

  f_ICMPv6_Start(IP, px_IPv6_RemoteAddress, cs_DrbInfo_EUTRA(eutra_CellI, tsc_DRB1));   /* @sic R5-106675 IP and IMS support sic@ */
  v_NasInd:= f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellI,

                  tsc_RRC_TI_Def,

                  cr_EstablishmentCause_Any,

                  cr_NAS_IndicationWithPiggybacking(v_AttReqSecurityHeader, //tsc_SHT_IntegrityProtected,
                                                    cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                    cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                              f_GetPdnType())));

    v_SecurityParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability(v_NasInd.Pdu.Msg.aTTACH_REQUEST);  // @sic R5s100135 sic@

    f_EUTRA_Security_Set(v_SecurityParams);

   if (v_NasInd.SecurityProtection.Status == tsc_SHT_NoSecurityProtection) // @sic R5s090210, R5s090373 sic@

    {

      v_ExpectedSecurityHeader := tsc_SHT_NoSecurityProtection;

    } else {

     v_ExpectedSecurityHeader := (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected);

    }

   //v_PdnConnectivityRequest := v_NasInd.Pdu.PiggybackedPduList[0].Msg;

    // Step 5-6: Authentication

    v_SecurityParams:=f_EUTRA_NAS_Authentication(eutra_CellI,

                                             tsc_SRB1,

                                             v_SecurityParams,

                                             tsc_SHT_NoSecurityProtection,

                                             v_ExpectedSecurityHeader,       // @sic R5s090210, R5s090373 sic@

                                             v_PLMN);                        // AUTHENTICATION REQUEST & RESPONSE

    // Step 7-8: NAS Security

    v_NasCountUL := f_EUTRA_NAS_ActivateSecurity(eutra_CellI, v_SecurityParams);   // NAS SECURITY COMMAND & COMPLETE

    // Step 10-11: RRC Security Mode procedures

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellI, v_SecurityParams, v_NasCountUL); // RRC SECURITY COMMAND & COMPLETE

    f_EUTRA_Security_Set ( v_SecurityParams ); //Saving security parameters

    // Step 12-13 UE Capability check

    f_EUTRA_Capability (eutra_CellI, tsc_RRC_TI_Def); // This enquires for the eutra capability but does not check the contents of the message received

    v_EPS_TI := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // Use this value in SS responses
    SRB.send(cas_SRB1_RrcNasPdu_REQ ( eutra_CellI,

                                      cs_TimingInfo_Now,

                                      cs_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est(tsc_RRC_TI_Def,

                                                                                       v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                       v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                       v_AntennaInfo

                                                                                       ),

                                      cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cds_ATTACH_ACCEPT_t3402(v_AttachType,

                                                                                cs_TAIListNonConsecutive_lv (v_PLMN, {bit2oct(v_TAC)}), //cs_TrackingAreaIdListDef_lv,
                                                                                f_GutiParameters2MobileIdentity(tsc_IEI_Guti, f_EUTRA_CellInfo_GetGuti(eutra_CellI)),

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.aTTACH_REQUEST.addUpdateType),

                                                                                cs_GprsTimer_tv('17'O,'001'B, '00010'B)), //2 minutes
                                                        cs_508_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                                  v_EPS_TI, //tsc_PTI_Unassigned,

                                                                                                  cs_AccessPointName(px_AccessPointName),

                                                                                                  f_GetPDNAddress (f_CheckPCOforIPallocationViaNas (omit)),

                                                                                                  omit,

                                                                                                  omit,

                                                                                                  f_GetDefPCOforPCSCFAddress (v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions))))); // @sic R5-104703 sic@

    //Receive RRCConnectionReconfigurationComplete

  SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellI, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

  f_EUTRA_NAS_AttachComplete(eutra_CellI, tsc_EpsDefaultBearerId);

  f_EUTRA_RRC_ConnectionRelease ( eutra_CellI, tsc_RRC_TI_Def, other );
/////////////////////////////////////////////////////////////////

  f_EUTRA_RbEst_Def(eutra_CellI);

  f_EUTRA_TestBody_Set(true);

  fl_TC_9_2_2_2_14_Body();

  // Postamble

  f_EUTRA_TestBody_Set(false);

  f_EUTRA_Postamble(eutra_CellI, E2_CONNECTED);
}


4.3 Change 2

	Testcase name
	9.2.2.2.14

	Protocol
	EMM

	Function Name
	function fl_TC_9_2_2_2_14_Body()

	Reason for change
	1. Steps 5 to 16 to be performed on Cell I 

	Summary of change
	1. Cell A is replaced with Cell I for steps 5 to 16

	Source of change
	File : …\9_2\NAS_Detach.ttcn


Before:
	function fl_TC_9_2_2_2_14_Body() runs on EUTRA_PTC

{

  var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

  var ReleaseCause v_ReleaseCause := other;

  var NAS_MSG_Indication_Type v_NasInd;

  var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType();

  var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

  timer t_T3402 := f_EUTRA_SetTimerToleranceMin (eutra_CellI, nasTimer, 120.0 ); // @sic R5s102190 sic@

// ==========================================================================

  //@siclog "Step 1" siclog@

  //+  SS initiate Detach procedure with the Detach Type IE "re-attach not required" and no EMM cause IE included

  SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellI,

                               cs_TimingInfo_Now,

                               cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                              cs_508_DETACH_REQUEST_MT(cs_DetachType(tsc_DetachMT_NoReAttach), omit))));

  //+ The UE transmits a DETACH ACCEPT message on Cell I. Note: Now UE should start timer T3402

  //@siclog "Step 2" siclog@

  SRB.receive(car_SRB2_NasPdu_IND(eutra_CellI,

                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                  cr_508_DETACH_ACCEPT)));

  t_T3402.start;

  //+The SS releases the RRC connection.

    //@siclog "Step 3" siclog@

  f_EUTRA_RRC_ConnectionRelease ( eutra_CellI, v_RRC_TI, v_ReleaseCause );

 // Check: When the timer T3402 expires does the UE transmit ATTACH REQUEST message on cell I?

   //@siclog "Step 4" siclog@

  t_T3402.timeout;

  v_NasInd:= f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellI,

                                            v_RRC_TI,

                                            cr_EstablishmentCause_Any,

                                            cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,

                                                                              cdr_ATTACH_REQUEST_WithIMSIAndnoTAI(v_AttachType,

                                                                                                                  v_AdditionalUpdateType,

                                                                                                                  f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                                                                  cr_TMSI_Status('0'B) ifpresent),

                                                                              cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                              f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    //@siclog "StepS 5 - 16" siclog@

    f_EUTRA_IdleUpdated_Step5_17(eutra_CellA, STATE2_IDLEUPDATE, NORMAL, v_NasInd, false);
    }// function tc_9_2_2_2_14_EUTRA


After:
	function fl_TC_9_2_2_2_14_Body() runs on EUTRA_PTC

{

  var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

  var ReleaseCause v_ReleaseCause := other;

  var NAS_MSG_Indication_Type v_NasInd;

  var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType();

  var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

  timer t_T3402 := f_EUTRA_SetTimerToleranceMin (eutra_CellI, nasTimer, 120.0 ); // @sic R5s102190 sic@

// ==========================================================================

  //@siclog "Step 1" siclog@

  //+  SS initiate Detach procedure with the Detach Type IE "re-attach not required" and no EMM cause IE included

  SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellI,

                               cs_TimingInfo_Now,

                               cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                              cs_508_DETACH_REQUEST_MT(cs_DetachType(tsc_DetachMT_NoReAttach), omit))));

  //+ The UE transmits a DETACH ACCEPT message on Cell I. Note: Now UE should start timer T3402

  //@siclog "Step 2" siclog@

  SRB.receive(car_SRB2_NasPdu_IND(eutra_CellI,

                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                  cr_508_DETACH_ACCEPT)));

  t_T3402.start;

  //+The SS releases the RRC connection.

    //@siclog "Step 3" siclog@

  f_EUTRA_RRC_ConnectionRelease ( eutra_CellI, v_RRC_TI, v_ReleaseCause );

 // Check: When the timer T3402 expires does the UE transmit ATTACH REQUEST message on cell I?

   //@siclog "Step 4" siclog@

  t_T3402.timeout;

  v_NasInd:= f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellI,

                                            v_RRC_TI,

                                            cr_EstablishmentCause_Any,

                                            cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,

                                                                              cdr_ATTACH_REQUEST_WithIMSIAndnoTAI(v_AttachType,

                                                                                                                  v_AdditionalUpdateType,

                                                                                                                  f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                                                                  cr_TMSI_Status('0'B) ifpresent),

                                                                              cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                              f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    //@siclog "StepS 5 - 16" siclog@

    f_EUTRA_IdleUpdated_Step5_17(eutra_CellI, STATE2_IDLEUPDATE, NORMAL, v_NasInd, false);
    }// function tc_9_2_2_2_14_EUTRA


4.4 Change 3

	Testcase name
	9.2.2.2.14

	Protocol
	EMM

	Function Name
	testcase TC_9_2_2_2_14()

	Reason for change
	UE takes up to a maximum of 300sec to attach on the VPLMN which is set as the serving cell in this test case. So, the existing guard timer value is not sufficient

	Summary of change
	300sec added to the current guard timer value.

	Source of change
	File : …\ LTE_EPS_TS_Testcases.ttcn


Before:
	testcase TC_9_2_2_2_14() runs on MTC system SYSTEM {

    // @purpose

    //   NW initiated detach / Abnormal case / EMM cause not included

    var EUTRA_PTC   v_EUTRA     := EUTRA_PTC.create alive;

    var UTRAN_PTC   v_UTRAN     := null;

    var GERAN_PTC   v_GERAN     := null;

    var CDMA2000_PTC v_CDMA2000  := null;

    timer t_GuardTimer := int2float(360); 
    f_MTC_ConnectPTCs(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000);

    v_EUTRA.start   (f_TC_9_2_2_2_14_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }


After:
	testcase TC_9_2_2_2_14() runs on MTC system SYSTEM {

    // @purpose

    //   NW initiated detach / Abnormal case / EMM cause not included

    var EUTRA_PTC   v_EUTRA     := EUTRA_PTC.create alive;

    var UTRAN_PTC   v_UTRAN     := null;

    var GERAN_PTC   v_GERAN     := null;

    var CDMA2000_PTC v_CDMA2000  := null;

    timer t_GuardTimer := int2float(660);  

    f_MTC_ConnectPTCs(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000);

    v_EUTRA.start   (f_TC_9_2_2_2_14_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }


5 Execution Log Files

5.1 LG L2000 UE

The LG L2000 UE passed these test cases on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files tc_9_2_2_2_14_LG-log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

5.2 Huawei Polaris UE

The Huawei Polaris UE passed these test cases on the Anite Conformance Toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files tc_9_2_2_2_14_Huawei-log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6 References

	[1]
	R5s100781: This archive comprises html format execution log file.

