
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of EMM test case 9.2.2.1.9 which are part of the LTE test suite.

The test case can be demonstrated to run with two LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Cases:
tc_9_2_2_1_9
Test Group:
EMM

ATS Version:
iwd-EUTRA-TVB2009-12_D10wk49
System Simulator used:
Anite Conformance Toolset
UE(s) used:
 Samsung Tanum and LG L2000 UE
Verification Status:
PASS
4 Corrections required for test case 9.2.2.1.9
4.1 Introduction

This section describes the changes required to make test case 9.2.2.1.9 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D10wk49 release.

4.2 Change 1

	Testcase name
	9.2.2.1.9

	Protocol
	EMM

	Function Name
	f_TC_9_2_2_1_9_EUTRA() 

	Reason for change
	1) Cells were wrongly initialized on different PLMNs
2) UE should be in state 3, generic RB established to make the test case more similar to real network scenario. 

	Summary of change
	1) Cells now initialized on the same PLMN
2) UE state moved to state 3 prior to entering the test body. 
NOTE : Prose changes will be submitted in the next RAN5 meeting

	Source of change
	NAS_Detach.ttcn


Before:
	//   **************************************************************

//                              TC_9_2_2_1_9

//   **************************************************************

function f_TC_9_2_2_1_9_EUTRA() runs on EUTRA_PTC

{

   // @desc UE initiated detach / Abnormal case / Change of cell into a new tracking area

   /* Test Purpose (TP)

    (1)

    with { UE in EMM-DEREGISTERED-INITIATED state }

    ensure that {

      when { the UE changes into a new tracking area that is not in the stored TAI list }

       then { the UE aborts the detach procedure and initiates a Tracking Area Updating procedure }

    }

    (2)

    with { UE in EMM-TRACKING-AREA-UPDATING-INITIATED state }

    ensure that {

      when { the UE receives TRACKING AREA UPDATE ACCEPT message }

       then { the UE re-initiates the detach procedure after completing the Tracking Area Updating procedure }

    }

    System Simulator:

        -   Cell A and Cell B are configured according to Table 6.3.2.2-1 in TS 36.508.

        -   Cell A belongs to TAI-1 (home PLMN) and is set to ''Serving cell''

        -   Cell B belongs to TAI-2 (home PLMN, another TAC) and is set to ''Non-Suitable cell''

    UE:

        None.

    Preamble:

        -   The UE is in state Registered, Idle mode (State 2) according to TS 36.508.

        -   UE has a valid GUTI-1 and is registered in TAI-1

    */

    var template (value) CellPowerList_Type v_CellPowerList;

    f_EUTRA_NAS_Init(LTE_CellsOnDifferentPLMN ,c3);

    // create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_EUTRA_CellConfig_Def(eutra_CellB);

    //Go to state 2

    f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE);

    // The SS pre-configures Cell A as the "Serving cell" and Cell B as "Non-Suitable cell"

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_ServingCellRS_EPRE ),

      cs_CellPower ( eutra_CellB, tsc_NonSuitableCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_2_1_9_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellB, E4_DEREGISTERED);

} // end f_TC_9_2_2_1_9_EUTRA


After:
	//   **************************************************************

//                              TC_9_2_2_1_9

//   **************************************************************

function f_TC_9_2_2_1_9_EUTRA() runs on EUTRA_PTC

{

   // @desc UE initiated detach / Abnormal case / Change of cell into a new tracking area

   /* Test Purpose (TP)

    (1)

    with { UE in EMM-DEREGISTERED-INITIATED state }

    ensure that {

      when { the UE changes into a new tracking area that is not in the stored TAI list }

       then { the UE aborts the detach procedure and initiates a Tracking Area Updating procedure }

    }

    (2)

    with { UE in EMM-TRACKING-AREA-UPDATING-INITIATED state }

    ensure that {

      when { the UE receives TRACKING AREA UPDATE ACCEPT message }

       then { the UE re-initiates the detach procedure after completing the Tracking Area Updating procedure }

    }

    System Simulator:

        -   Cell A and Cell B are configured according to Table 6.3.2.2-1 in TS 36.508.

        -   Cell A belongs to TAI-1 (home PLMN) and is set to ''Serving cell''

        -   Cell B belongs to TAI-2 (home PLMN, another TAC) and is set to ''Non-Suitable cell''

    UE:

        None.

    Preamble:

        -   The UE is in state Registered, Idle mode (State 2) according to TS 36.508.

        -   UE has a valid GUTI-1 and is registered in TAI-1

    */

    var template (value) CellPowerList_Type v_CellPowerList;

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN ,c3);

    // create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_EUTRA_CellConfig_Def(eutra_CellB);

    //Go to state 2

    f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE);

    //Go to state 3

    f_EUTRA_RbEst_Def ( eutra_CellA ); 
    // The SS pre-configures Cell A as the "Serving cell" and Cell B as "Non-Suitable cell"

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_ServingCellRS_EPRE ),

      cs_CellPower ( eutra_CellB, tsc_NonSuitableCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_2_1_9_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellB, E4_DEREGISTERED);

} // end f_TC_9_2_2_1_9_EUTRA


4.3 Change 2

	Testcase name
	9.2.2.1.9

	Protocol
	EMM

	Function Name
	fl_TC_9_2_2_1_9_Body() 

	Reason for change
	1) At step 2, since the UE has SRB2 established, UE will transmit detach request on SRB2 without any need for rrc Connection establishment. 
2) At step 4, cell power level should be changed to suitable neighbour cell since UE should receive the subsequent rrcConnectionReconfiguration message without any error 

3) After step 4, UE will loose the RRC Connection and as per 24.301 section 5.5.2.4 clause b, UE would enter EMM-DEREGISTERED state if the lower layer failure occurs before reception of detach accept. Due to this, UE does not trigger TAU request on cell B. Hence, for the UE to detect tracking area change, the SS should initiate handover of the UE to a cell of different tracking area after reception of detach request.
4) After handover, since the UE has SRB2 established, UE will transmit TAU request on SRB2 at step 5 without any need for rrc Connection establishment.

5) At step 5, as per 24.301, 5.5.3.2.2 , if a UE has a pending uplink signalling not related to the tracking area updating procedure, it may also set an "active" flag in the TRACKING AREA UPDATE REQUEST message to indicate the request to establish the user plane to the network and to keep the NAS signalling connection after the completion of the tracking area updating procedure. In this test case, at step 8, since Detach Request is pending, some UEs may set the active flag while sending TAU Request at step 5. Since this is UE implementation dependent, the test case should be able to handle any allowed value of active flag sent by UE 
6)  At step  6, SS should transmit TAU Accept on SRB2 since UE has SRB2 established

7) At step 7, since the UE has SRB2 established, UE will transmit TAU Complete on SRB2 

8)  At step 8, since the UE has SRB2 established, UE will transmit detach request on SRB2 without any need for rrc Connection establishment
9) At step 9, SS will transmit detach accept on SRB2. 

10) Connection release should be done on Cell B instead of cell L.



	Summary of change
	1) At step 2, ttcn changed to accept detach request on SRB 2. . 

2) At step 4, cell power level changed to suitable neighbour cell 

3) After step 4, handover initiated to cellB
4) At step 5, ttcn changed to accept TAU request on SRB2
5) At step 5, new template of TAU Request used to accept any allowed value of active flag in the TAU request message. 
6) At step  6, SS transmits TAU Accept on SRB2 
7) At step 7, SS receives TAU Complete on SRB2 

8) At step 8, SS receives detach request on SRB2 
9) At step 9, SS transmits detach accept on SRB2. 

10) Cell B is used for connection release instead of cell L


	Source of change
	NAS_Detach.ttcn , EUTRA_NASCommonTemplates.ttcn


Before:
	function fl_TC_9_2_2_1_9_Body() runs on EUTRA_PTC

{

    var NAS_MSG_Indication_Type v_NasInd;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var template (value) CellPowerList_Type v_CellPowerList;

    var GutiParameters_Type v_GutiParams1 := f_EUTRA_CellInfo_GetGuti (eutra_CellA);

    var GutiParameters_Type v_GutiParams_B := f_EUTRA_CellInfo_GetGuti (eutra_CellB);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams_B.PLMN_Identity);

    var template (omit) EPS_BearerContextStatus v_EpsBearerCtxtStatus;

    var float v_NoReactionTime := f_EUTRA_SetTimerToleranceMax( eutra_CellB, rrcTimer, 3.0 );

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC ( eutra_CellB );

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode ( eutra_CellB );

    var template (omit) LocAreaId v_LAI := f_GetLAI ( v_PLMN, v_LAC);

    var template (omit) MobileIdentity v_MSId := f_GetMSId ( );

    var template (value) TrackingAreaIdList v_TaiList := cds_TAIListNonConsecutive_tlv ( v_PLMN, { bit2oct(v_Tac) } );

    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType();

    //  Cause UE to initiate detach from the EPS services.

    //@siclog "Step 1" siclog@

    f_UT_InitiateDetach_EPSOnly(UT);

    //       The UE transmits a DETACH REQUEST message with the Detach type IE indicating

    //       'EPS detach' and 'normal detach'? The UE starts timer T3421.

    //@siclog "Step 2" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_MoSignalling,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cr_DETACH_REQUEST_NormalDetachandEpsOnly (cr_NAS_KeySetIdentifier (?, tsc_NasKsi_NativeSecurityContext),

                                                                                                            f_GutiParameters2MobileIdentity(omit, v_GutiParams1)

                                               )));
    // The SS does not respond to the DETACH REQUEST message.

    // Set the cell type of cell A to the "Non-Suitable cell".

    //         Set the cell type of cell B to the "Serving cell".

    // Note: T3421 value is specified as 15s in TS 24.301 and it is assumed that

    //       SS can configure cells within this time.

    //@siclog "Step 3 - 4" siclog@

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_NonSuitableCellRS_EPRE ),

      cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

    //  Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message?

    //@siclog "Step 5" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_Any,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_TAU_Request(v_UpdateType,

                                                                                    f_EUTRA_SecurityKSIasme_Get(),

                                                                                    v_AdditionalUpdateType

                                                                                    )));
    v_EpsBearerCtxtStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5: TAU Request sent by UE");

    // The SS transmits a TRACKING AREA UPDATE ACCEPT message.

    //@siclog "Step 6" siclog@

    SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_CellB,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_508_TAU_Accept ( v_UpdateType,

                                                                          f_GutiParameters2MobileIdentity (tsc_IEI_Guti, v_GutiParams_B ), //GUTI-2

                                                                          v_TaiList,

                                                                          v_EpsBearerCtxtStatus,

                                                                          v_LAI,

                                                                          v_MSId,

                                                                          f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));

    // Check: Does the UE transmit a TRACKING AREA UPDATE COMPLETE message?

    //@siclog "Step 7" siclog@

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete)

                                   ));

    //  The UE transmits a DETACH REQUEST message with the Detach type IE indicating

    //         'EPS detach' and 'normal detach'? The UE starts timer T3421.

    //@siclog "Step 8" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_MoSignalling,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cr_DETACH_REQUEST_NormalDetachandEpsOnly (cr_NAS_KeySetIdentifier (?, tsc_NasKsi_NativeSecurityContext),

                                                                                                            f_GutiParameters2MobileIdentity(omit, v_GutiParams_B)

                                               )));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8: Detach Request sent by UE");

    // SS responds with DETACH ACCEPT message

    //@siclog "Step 9" siclog@

    SRB.send (cas_SRB1_NasPdu_REQ ( eutra_CellB,

                                    cs_TimingInfo_Now,

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DETACH_ACCEPT)

                                  ));

    f_EUTRA_RRC_ConnectionRelease(eutra_CellL, tsc_RRC_TI_Def, other);

     //  Check: Does the test result of generic procedure in TS 36.508 subclause 6.4.2.5 indicate that

     // the UE does not respond to paging when paged with GUTI-2 and with CN domain indicator set to "PS"?

     //@siclog "Step 10" siclog@

     f_EUTRA_508Check_NASNoResponse ( eutra_CellB,

                                      cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_GutiParams_B.MME_Code, v_GutiParams_B.M_TMSI )),

                                      v_NoReactionTime );

}// end function fl_TC_9_2_2_1_9_Body


After:
	NAS_Detach.ttcn

-----------------------------------------------------------------------------

function fl_TC_9_2_2_1_9_Body() runs on EUTRA_PTC

{

    var NAS_MSG_Indication_Type v_NasInd;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var template (value) CellPowerList_Type v_CellPowerList;

    var GutiParameters_Type v_GutiParams1 := f_EUTRA_CellInfo_GetGuti (eutra_CellA);

    var GutiParameters_Type v_GutiParams_B := f_EUTRA_CellInfo_GetGuti (eutra_CellB);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams_B.PLMN_Identity);

    var template (omit) EPS_BearerContextStatus v_EpsBearerCtxtStatus;

    var float v_NoReactionTime := f_EUTRA_SetTimerToleranceMax( eutra_CellB, rrcTimer, 3.0 );

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC ( eutra_CellB );

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode ( eutra_CellB );

    var template (omit) LocAreaId v_LAI := f_GetLAI ( v_PLMN, v_LAC);

    var template (omit) MobileIdentity v_MSId := f_GetMSId ( );

    var template (value) TrackingAreaIdList v_TaiList := cds_TAIListNonConsecutive_tlv ( v_PLMN, { bit2oct(v_Tac) } );

    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType();

    var SRB_COMMON_IND v_ReceivedAsp; 

    //  Cause UE to initiate detach from the EPS services.

    //@siclog "Step 1" siclog@

    f_UT_InitiateDetach_EPSOnly(UT);

    //       The UE transmits a DETACH REQUEST message with the Detach type IE indicating

    //       'EPS detach' and 'normal detach'? The UE starts timer T3421.

    //@siclog "Step 2" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication (tsc_SHT_IntegrityProtected_Ciphered,

                                          cr_DETACH_REQUEST_NormalDetachandEpsOnly (cr_NAS_KeySetIdentifier (?, tsc_NasKsi_NativeSecurityContext),

                                                 f_GutiParameters2MobileIdentity(omit, v_GutiParams1)

                                   ))));
    // Set the cell type of cell A to the "Non-Suitable cell".

    //         Set the cell type of cell B to the "Serving cell".

    // Note: T3421 value is specified as 15s in TS 24.301 and it is assumed that

    //       SS can configure cells within this time.

    //@siclog "Step 3 - 4" siclog@

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),

      cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

    //SS now creates a handover situation to direct the UE to cellB

    f_EUTRA_508RRC_IntraLTE_HO_InterCell ( eutra_CellA, eutra_CellB ); 

    //Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message?

    //@siclog "Step 5" siclog@

    SRB.receive( car_SRB2_NasPdu_IND(eutra_CellB,

                                     cr_NAS_Indication(

                                     tsc_SHT_IntegrityProtected,

                                     cdr_TAU_Request_92219(

                                                       v_UpdateType,

                                                       f_EUTRA_SecurityKSIasme_Get(),

                                                       v_AdditionalUpdateType))))

            -> value v_ReceivedAsp;

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0] ;
    v_EpsBearerCtxtStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;        

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5: TAU Request sent by UE");

    // The SS transmits a TRACKING AREA UPDATE ACCEPT message.

    //@siclog "Step 6" siclog@

    SRB.send ( cas_SRB2_NasPdu_REQ ( eutra_CellB,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered, 

                                                      cs_508_TAU_Accept ( v_UpdateType,

                                                                          f_GutiParameters2MobileIdentity (tsc_IEI_Guti, v_GutiParams_B ), //GUTI-2

                                                                          v_TaiList,

                                                                          v_EpsBearerCtxtStatus,

                                                                          v_LAI,

                                                                          v_MSId,

                                                                          f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));

    // Check: Does the UE transmit a TRACKING AREA UPDATE COMPLETE message?

    //@siclog "Step 7" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete)

                                   ));

    //  The UE transmits a DETACH REQUEST message with the Detach type IE indicating

    //         'EPS detach' and 'normal detach'? The UE starts timer T3421.

    //@siclog "Step 8" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,

                                    cr_NAS_Indication (tsc_SHT_IntegrityProtected_Ciphered,

                                          cr_DETACH_REQUEST_NormalDetachandEpsOnly (cr_NAS_KeySetIdentifier (?, tsc_NasKsi_NativeSecurityContext),

                                                 f_GutiParameters2MobileIdentity(omit, v_GutiParams_B)

                                   ))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8: Detach Request sent by UE");

    // SS responds with DETACH ACCEPT message

    //@siclog "Step 9" siclog@

    SRB.send (cas_SRB2_NasPdu_REQ ( eutra_CellB,

                                    cs_TimingInfo_Now,

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DETACH_ACCEPT)

                                  ));

    f_EUTRA_RRC_ConnectionRelease(eutra_CellB, tsc_RRC_TI_Def, other);

     //  Check: Does the test result of generic procedure in TS 36.508 subclause 6.4.2.5 indicate that

     // the UE does not respond to paging when paged with GUTI-2 and with CN domain indicator set to "PS"?

     //@siclog "Step 10" siclog@

     f_EUTRA_508Check_NASNoResponse ( eutra_CellB,

                                      cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_GutiParams_B.MME_Code, v_GutiParams_B.M_TMSI )),

                                      v_NoReactionTime );

}// end function fl_TC_9_2_2_1_9_Body

EUTRA_NASCommonTemplates.ttcn

---------------------------------------------------------------------------------------------------------------------

  template NAS_UL_Message_Type cdr_TAU_Request_92219(EPS_UpdateTypeValue p_EpsUpdate_TypeValue,

                                                                 NAS_KsiValue p_KsiValue,

                                                                 template AdditionalUpdateType p_AdditionalUpdateType

                                                                 )

    modifies cr_508_TAU_Request :=

  { 

    tRACKING_AREA_UPDATE_REQUEST := {

      epsUpdateType := cr_EPS_UpdateType(?, p_EpsUpdate_TypeValue)

    }

  };


5 Execution Log Files

5.1 Samsung Tanum UE

The Samsung Tanum UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_9_2_2_1_9_Samsung-LOG.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.
5.2 LG L2000 UE

The LG L2000 UE passed these test cases on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files tc_9_2_2_1_9_LG-log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6 References

	[1]
	R5s100884: This archive comprises html format execution log file.

